// BlinnPhong 光照模型 求高光反射
// 高光 = 高光颜色 * 高光系数 * pow(max(0,dot(世界法线,半角向量)),高光范围)
//specular = specularColor * ks * pow(max(0,dot(N,H)),shininess)
Shader "Study/Chapter5/BlinnPhong"
{
    Properties
    {
		  // 高光颜色
        _SpecularColor("SpecularColor",Color) = (1,1,1,1)
        // 高光系数
        _SpecularIntensity("SpecularIntensity",float) = 1
        // 高光范围
        _Shininess("Shininess",float) = 1
    }
    SubShader
    {
        Tags { "RenderType"="Opaque" }
        LOD 100

        Pass
        {
            CGPROGRAM
            #pragma vertex vert
            #pragma fragment frag

            #include "UnityCG.cginc"

            struct appdata
            {
                float4 vertex : POSITION;
				float3 normal : NORMAL;
            };

            struct v2f
            {
                float4 vertex : SV_POSITION;
				float3 worldPos : TEXCOORD0;
				float3 worldNormal : TEXCOORD1;
            };

			float4 _SpecularColor;
			float _SpecularIntensity;
			float _Shininess;

            v2f vert (appdata v)
            {
                v2f o;
                o.vertex = UnityObjectToClipPos(v.vertex);
				o.worldPos = mul(unity_ObjectToWorld,v.vertex);
				o.worldNormal = UnityObjectToWorldNormal(v.normal);
                return o;
            }

            fixed4 frag (v2f i) : SV_Target
            {
				// 高光 = 高光颜色 * 高光系数 * pow(max(0,dot(世界法线,半角向量)),高光范围)
                //specular = specularColor * ks * pow(max(0,dot(N,H)),shininess)
				float3 N = normalize(i.worldNormal);   // 法线归一化
				float3 L = normalize(_WorldSpaceLightPos0.xyz);			// 入射光线
				float3 V = normalize(_WorldSpaceCameraPos.xyz - i.worldPos);   // 观察向量

				// 半角向量 H 即相当于入射光线和观察方向的中间向量
                float3 H = normalize(L+V);
				float NdotH = dot(N,H);

				fixed4 specular = _SpecularColor * _SpecularIntensity * pow(max(0,NdotH),_Shininess);

                return specular;
            }
            ENDCG
        }
    }
}
